// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/structs/EnumerableSet.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/structs/EnumerableSet.sol)
// This file was procedurally generated from scripts/generate/templates/EnumerableSet.js.

pragma solidity ^0.8.20;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```solidity
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 *
 * [WARNING]
 * ====
 * Trying to delete such a structure from storage will likely result in data corruption, rendering the structure
 * unusable.
 * See https://github.com/ethereum/solidity/pull/11843[ethereum/solidity#11843] for more info.
 *
 * In order to clean an EnumerableSet, you can either remove all elements one by one or create a fresh instance using an
 * array of EnumerableSet.
 * ====
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;
        // Position is the index of the value in the `values` array plus 1.
        // Position 0 is used to mean a value is not in the set.
        mapping(bytes32 value => uint256) _positions;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._positions[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We cache the value's position to prevent multiple reads from the same storage slot
        uint256 position = set._positions[value];

        if (position != 0) {
            // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 valueIndex = position - 1;
            uint256 lastIndex = set._values.length - 1;

            if (valueIndex != lastIndex) {
                bytes32 lastValue = set._values[lastIndex];

                // Move the lastValue to the index where the value to delete is
                set._values[valueIndex] = lastValue;
                // Update the tracked position of the lastValue (that was just moved)
                set._positions[lastValue] = position;
            }

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the tracked position for the deleted slot
            delete set._positions[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._positions[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        return set._values[index];
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function _values(Set storage set) private view returns (bytes32[] memory) {
        return set._values;
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(Bytes32Set storage set) internal view returns (bytes32[] memory) {
        bytes32[] memory store = _values(set._inner);
        bytes32[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint160(uint256(_at(set._inner, index))));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(AddressSet storage set) internal view returns (address[] memory) {
        bytes32[] memory store = _values(set._inner);
        address[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(UintSet storage set) internal view returns (uint256[] memory) {
        bytes32[] memory store = _values(set._inner);
        uint256[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }
}

// File: contracts/HoodyGangStake.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.19;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/structs/EnumerableSet.sol";
import "./IHoody.sol";
import "./IHoodySign.sol";

interface IHoodyGang is IHoody {
    function transferFrom(address, address, uint256) external;

    function ownerOf(uint256) external view returns (address);

    function approve(address, uint256) external;
}

interface IHoodyCredit {
    function transfer(address, uint256) external returns (bool);

    function balanceOf(address) external returns (uint256);
}

contract HoodyGangStake is IHoody, Ownable {
    using EnumerableSet for EnumerableSet.UintSet;

    uint256 public rewardInterval = 1 days;
    mapping(address => EnumerableSet.UintSet) stakedNFTsByHolder;
    mapping(HoodyGangRarity => uint256) public rewardPointByRarity;
    mapping(uint256 => HoodyGangRarity) public stakedNFTsRarity;
    mapping(uint256 => uint256) public lastClaimedTime;

    address public hoodyGang;
    address public hoodyCredit;
    address public hoodySign;

    bool public rewardAvailable = true;

    event StakeNFT(address indexed holder, uint256 tokenID);
    event UnStakeNFT(address indexed holder, uint256 tokenID);
    event ResetTimer(uint256 tokenID, uint256 claimTime);

    constructor(address _hoodyGang, address _hoodyCredit) Ownable(msg.sender) {
        hoodyGang = _hoodyGang;
        hoodyCredit = _hoodyCredit;
    }

    function stake(
        uint256[] memory _tokenIds,
        HoodyGangRarity[] memory _rarities,
        bytes memory _signature
    ) external {
        require(
            IHoodySign(hoodySign).verifyForStake(
                msg.sender,
                _tokenIds,
                _rarities,
                _signature
            ),
            "Invalid Signature"
        );

        IHoodySign(hoodySign).increaseNonce(msg.sender);

        for (uint i; i < _tokenIds.length; i++) {
            require(
                IHoodyGang(hoodyGang).ownerOf(_tokenIds[i]) == msg.sender,
                "Not your token!"
            );
            stakedNFTsRarity[_tokenIds[i]] = _rarities[i];
            stakedNFTsByHolder[msg.sender].add(_tokenIds[i]);
            IHoodyGang(hoodyGang).approve(address(this), _tokenIds[i]);
            IHoodyGang(hoodyGang).transferFrom(
                msg.sender,
                address(this),
                _tokenIds[i]
            );
            lastClaimedTime[_tokenIds[i]] = block.timestamp;

            emit StakeNFT(msg.sender, _tokenIds[i]);
        }
    }

    function unstake(uint256[] memory _tokenIds) external {
        claimCredit();
        for (uint i; i < _tokenIds.length; i++) {
            require(
                stakedNFTsByHolder[msg.sender].contains(_tokenIds[i]),
                "Not your token!"
            );
            stakedNFTsByHolder[msg.sender].remove(_tokenIds[i]);
            IHoodyGang(hoodyGang).transferFrom(
                address(this),
                msg.sender,
                _tokenIds[i]
            );

            emit UnStakeNFT(msg.sender, _tokenIds[i]);
        }
    }

    function calcTotalCreditsByHolder(
        address _holder
    ) public view returns (uint256) {
        uint256[] memory tokenIds = getStakeTokensByHolder(_holder);
        uint256 credit;
        for (uint256 i; i < tokenIds.length; i++) {
            uint256 tokenId = tokenIds[i];
            if (block.timestamp >= lastClaimedTime[tokenId] + rewardInterval) {
                credit +=
                    rewardPointByRarity[stakedNFTsRarity[tokenId]] *
                    ((block.timestamp - lastClaimedTime[tokenId]) /
                        rewardInterval);
            }
        }
        return credit;
    }

    function claimCredit() public {
        if (rewardAvailable) {
            uint256[] memory tokenIds = getStakeTokensByHolder(msg.sender);
            uint256 credit;
            for (uint256 i; i < tokenIds.length; i++) {
                uint256 tokenId = tokenIds[i];
                if (
                    block.timestamp >= lastClaimedTime[tokenId] + rewardInterval
                ) {
                    uint256 passed = (block.timestamp -
                        lastClaimedTime[tokenId]) / rewardInterval;
                    credit +=
                        rewardPointByRarity[stakedNFTsRarity[tokenId]] *
                        passed;
                    lastClaimedTime[tokenId] += passed * rewardInterval;
                }

                emit ResetTimer(tokenId, lastClaimedTime[tokenId]);
            }
            IHoodyCredit(hoodyCredit).transfer(msg.sender, credit);
        }
    }

    function getStakeTokensByHolder(
        address _holder
    ) public view returns (uint256[] memory) {
        uint256[] memory tokenIds = new uint256[](
            stakedNFTsByHolder[_holder].length()
        );
        for (uint256 i; i < stakedNFTsByHolder[_holder].length(); i++) {
            tokenIds[i] = stakedNFTsByHolder[_holder].at(i);
        }
        return tokenIds;
    }

    function getLastClaimedTime(
        uint256[] memory _tokenIds
    ) external view returns (uint256[] memory) {
        uint256[] memory claimTimes = new uint256[](_tokenIds.length);
        for (uint i; i < _tokenIds.length; i++) {
            claimTimes[i] = lastClaimedTime[_tokenIds[i]];
        }

        return claimTimes;
    }

    function setRewardPointByRarity(
        HoodyGangRarity[] memory _rarities,
        uint256[] memory _points
    ) external onlyOwner {
        require(_rarities.length == _points.length, "Invalid Params!");
        for (uint i; i < _rarities.length; i++) {
            rewardPointByRarity[_rarities[i]] = _points[i];
        }
    }

    function withdraw() external onlyOwner {
        uint256 balance = IHoodyCredit(hoodyCredit).balanceOf(address(this));
        IHoodyCredit(hoodyCredit).transfer(owner(), balance);
    }

    function closeStaking() external onlyOwner {
        rewardAvailable = false;
    }

    function setHoodySign(address _hoodySign) external onlyOwner {
        hoodySign = _hoodySign;
    }

    function setHoodyGang(address _hoodyGang) external onlyOwner {
        hoodyGang = _hoodyGang;
    }

    function setHoodyCredit(address _hoodyCredit) external onlyOwner {
        hoodyCredit = _hoodyCredit;
    }

    function setRewardInterval(uint256 _rewardInterval) external onlyOwner {
        rewardInterval = _rewardInterval;
    }
}

// File: contracts/IHoody.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.19;

interface IHoody {
    enum HoodyGangRarity {
        Common,
        Rare,
        Legendary
    }

    enum HoodyTraitRarity {
        Common,
        Rare,
        Legendary
    }
}

// File: contracts/IHoodySign.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.19;
import "./IHoody.sol";

interface IHoodySign is IHoody {
    function verifyForTraits(
        address,
        string memory,
        uint16[] memory,
        bytes memory
    ) external view returns (bool);

    function verifyForStake(
        address,
        uint256[] memory,
        HoodyGangRarity[] memory,
        bytes memory
    ) external view returns (bool);

    function verifyForMigrate(
        address,
        uint256,
        string memory,
        bytes memory
    ) external view returns (bool);

    function increaseNonce(address) external;
}